
Abstract 

Life expectancy (LE) tables are statistical tools typically used to portray the expectation of life at various 
ages and are a fundamental tool in survival analysis. They also provide information on the number of 
individuals who survive to various ages, median age at death, age-specific death rates, and the 
probability of dying at certain ages. LE at birth is the most frequently cited LE table statistic. The Texas 
Department of State Health Services (DSHS) creates LE for publication in the Vital Statistics Annual 
Report every year. Previously, these tables were constructed in a multi-step process using detailed 
Python and Java script. In this paper we provide a general framework for building LE tables using SAS.  

Introduction 

 
The LE table has become one of the most popular statistics in community health assessment. Like Infant 
Mortality Rates, they help to take a ‘snapshot’ of the health of a community or population. LE tables are 
typically used to portray the expectation of life in specific areas or populations at various ages. At the 
same time, LE tables provide information on survivals to various ages, the median age at death, age-
specific death rates, and the probability of dying at certain ages. Life expectancy at birth simply 
constitutes our best estimate of how long a person might live if medical practices remain the same.1 The 
Texas DSHS often get requests from our clients to “personalize” the LE table to their specific geographic 
areas, like Texas counties or by health regions. The time and resources involved in making these tables2 
are currently prohibitive. 
 
For the past twenty years, the Texas DSHS, Center for Health Statistics, have used several computer 
programs (a combination of Survival 5.0, Java, Python, and Microsoft Excel) to generate LE tables.3 To 
replace this detailed process, we searched the existing SAS papers for scripts to build LE tables but could 
not find one that suited our needs.4 Neither PROC LIFETEST nor PROC LIFEREG produced the output 
we desired. We therefore developed a SAS syntax which constructs LE tables in a few simple steps. 
 
In the sections below, we describe the calculation method and the SAS syntax which generates a LE 
table. Two techniques which are used here are the LAG() function and the RETAIN statement.  
 

Getting Started: The Mathematics1 

There are several columns which make up the life tables. Getting familiar with them will help us to 
understand how the life table is built. These parameters are:  
 

1) Age interval (x to x+n): the period of life between two exact ages. Here x indicates the starting 
point for an age interval, and n is the interval length. 

 
2) Proportion dying (qx): the proportion of persons alive at the beginning of each age interval who 

die before reaching the end of the age interval.  
 

a. For infants, q0 is calculated as: 



 

𝑞0 =
𝐼𝑛𝑓𝑎𝑛𝑡 𝐷𝑒𝑎𝑡ℎ𝑠

𝑈𝑛𝑤𝑒𝑖𝑔ℎ𝑡𝑒𝑑 𝐿𝑖𝑣𝑒 𝐵𝑖𝑟𝑡ℎ𝑠
 

 
b. For age group of 75+, we set the 𝑞75 = 1 because this cohort will all die. 

 
c. For all other age cohorts, qx is calculated as: 

   

𝑞𝑥 =
2 ∗ 𝑛 ∗ 𝑑𝑟

(2 + 𝑛 ∗ 𝑑𝑟)
 

 
              Where n is the number of years within an age interval and dr is the age-specific death rate.  
 

Proportion surviving (px): the proportion of persons alive at the beginning of each age interval who 
survive over the age interval. 

𝑝𝑥 = 1 − 𝑞𝑥 
 

3) Number surviving (lx): number of people living at the beginning of each age interval.  
Life table usually represents a hypothetical cohort of 100,000 persons born at the same instant, 
therefore l0 = 100,000. For the other age cohorts,  
 

𝐼𝑥 = 𝐼𝑥−𝑛 ∗ 𝑝𝑥−𝑛  

 
Number dying (dx): number of persons dying during the age interval.  
 

𝑑𝑥 = 𝐼𝑥 − 𝐼𝑥+𝑛 
 

4) The number of person-years lived in each age interval:  
 

𝐿𝑥 = 𝑛 ∗ 𝐼𝑥+𝑛 + 𝑑𝑥 ∗ 𝑎𝑥 
 

             Here ax is the average amount of time lived in interval x to x+n by those dying in the interval.  
 

5) The number of person-years lived in each age interval and all subsequent age intervals:   
 

𝑇𝑥 = 𝐿𝑥 + 𝐿𝑥+𝑛 + ⋯ + 𝐿𝑤 

𝑜𝑟 

𝑇𝑥 = 𝑇𝑥+𝑛 + 𝐿𝑥 
 

Average remaining life time (ex) or the expectation of life at any given age (the average years 
remaining to be lived by those surviving to that age). 
 

𝑒𝑥 =
𝑇𝑥

𝐼𝑥

 

 
  



Translating from Formulas to SAS Syntax 

We will use the following data to generate a related life table. The numbers are from Texas Vital Statistics 
Annual Report, Table 24 for the reporting year of 20145.  
  

Table 1. Dataset used to generate Life Table for Texas Residents, 2014 

SexRace years agegroup population deaths births ax 
Texas Total 1 0 406,481 2,270 403,439 0.15 

Texas Total 4 1 1,581,709 423 403,439 1.65 

Texas Total 5 5 1,972,064 260 403,439 2.25 

Texas Total 5 10 2,048,295 281 403,439 3.05 

Texas Total 5 15 2,040,335 929 403,439 2.75 

Texas Total 5 20 2,006,970 1,733 403,439 2.55 

Texas Total 5 25 1,940,844 1,912 403,439 2.5 

Texas Total 5 30 1,976,615 2,336 403,439 2.55 

Texas Total 5 35 1,865,823 2,677 403,439 2.65 

Texas Total 5 40 1,830,349 3,469 403,439 2.7 

Texas Total 5 45 1,742,003 5,233 403,439 2.7 

Texas Total 5 50 1,765,609 8,712 403,439 2.65 

Texas Total 5 55 1,673,253 12,418 403,439 2.65 

Texas Total 5 60 1,403,713 15,151 403,439 2.65 

Texas Total 5 65 1,143,547 17,579 403,439 2.6 

Texas Total 5 70 798,713 18,986 403,439 2.6 

Texas Total 25 75 1,272,791 94,788 403,439 11.5 

 
1) We imported the data into SAS® 9.4 and named the dataset ‘test’. 
2) We then calculated three new variables:  

a. qx, the proportion of persons alive at the beginning of each age interval who die before 
reaching the end of the age interval;  

b. dr, age specific death rate; and 
c. px, the probability of surviving over the age interval.  

 
For age group 0, we used the actual number of births in the calculation of  

q0: 𝑞0 =
𝐼𝑛𝑓𝑎𝑛𝑡 𝐷𝑒𝑎𝑡ℎ𝑠

𝑈𝑛𝑤𝑒𝑖𝑔ℎ𝑡𝑒𝑑 𝐿𝑖𝑣𝑒 𝐵𝑖𝑟𝑡ℎ𝑠
. For age group 75+, we set q75 to 1.0 since the cohort will eventually 

all die. For all other age groups, qx was calculated as 𝑞𝑥 =
2∗𝑛∗𝑑𝑟

(2+𝑛∗𝑑𝑟)
, where n is the column ‘years’ 

and dr the age-specific death rate. 
 

Data test1;  

   Set test; 

   Length qx dr px 8; 

   If agegroup=0 then Do; 

     dr=deaths/births; 

     qx=deaths/births; 

   End; 

   Else Do; 

     dr=deaths/population; 

     If agegroup=1 then qx=2*4*dr/(2+4*dr); 

  Else If agegroup=75 then qx=1.000000; 

     Else qx=2*5*dr/(2+5*dr); 

     End; 

   px=1-qx; 

   Format qx dr px 8.5; 

Run; 

 

3) In the next step, we calculated the number surviving (Ix), which denotes the number of people 
living at the beginning of each age interval. The first number of this column, I0, is an arbitrary 



number which is usually set to 100,000, meaning 100,000 live birth happening at the same 
instant. Each successive number represents the number of survivors at the age(x). To calculate 
the probability of surviving at age(x), we create two new variables, x and xx. x is initially used to 
hold the value of p0. In each of the following age group, the value of x is passed to xx using a 
retain statement, and x itself is used to calculate the survival probability of the following age 
group. The surviving population at the beginning of each age interval is calculated as 𝐼𝑥 =
100,000 ∗ 𝑥𝑥. 
 

Data test2; 

   Set test1; 

   Length x xx Ix 8; 

   If years=1 then Do; 

        x=1; 

   x=x*px; 

   xx=1; 

   Ix=100000; 

   End; 

   Else do; 

      Retain x; 

        xx=x; 

   Ix=100000*xx; 

   x=x*px;  

   End; 

Format Ix comma8.0; 

run; 

 

4) In order to calculate the expected number of deaths in each age group (dx) and number of 
person- years lived(Lx), we first inverted the dataset so that we can use the LAG() function to 
capture the previous Ix, which is the number of people surviving at the beginning of the next age 
group. The expected number of deaths in each age group, dx, was calculated as 𝑑𝑥 =  𝐼𝑥 −
𝐿𝑎𝑔(𝐼𝑥). The variable LagIx is set to 0 for the last age group, 75 years and older, because this 
population will stay until death.  
 
Each of the dx people who die during the interval x and x+n has lived x complete years plus some 
fraction of the years n. The average of these fractions, denoted by ax, is usually set as a constant 
and we have these values pre-determined in the table.  
 
Since each member of the cohort who survives the year interval x to x+n contributes “n” number 
of years to Lx. While those members who die during this period of time contributes, on the 
average, a fraction 𝑎𝑥 of the “n” number of years, so that 𝐿𝑥 = 𝑛 ∗ 𝐿𝑎𝑔I𝑥 + 𝑑𝑥 ∗ 𝑎𝑥. 
 

Proc Sort Data=test2; 

   by descending Agegroup; 

Run; 

 

Data test3; 

   Set test2; 

   Length LagIx dx Lx 8; 

   LagIx=lag(Ix); 

   if Agegroup=75 then LagIx=0; 

   dx = Ix -LagIx;  

   Lx=years*LagIx + dx*ax; 

   Format LagIx dx Lx comma8.0; 

Run; 

 



5) Now we were very close to getting the life expectancies, but there was still one more step. We 
calculated the total number of years lived beyond age(x), Tx. This is equal to the sum of the 
number of years lived in each age interval beginning with age x, and can be calculated as 𝑇𝑥 =
𝐿𝑥 +  𝐿𝑥+𝑛 +  … + 𝐿75, or 𝑇𝑥 =  𝐿𝑥  +  𝑇𝑥+𝑛 . After we get Tx, the number of years yet to be lived by 

a person now at age x is calculated as 𝑒𝑥 =
𝑇𝑥

𝐼𝑥
 . 

 
Now that we have completed all calculations, the table was inverted to the original order from the 
youngest age to the oldest.  
 

Data Test4;  

  Tx = 0; 

    Do i = 1 to 17; 

    Set Test3;  

    Tx + Lx; 

    Ex=Tx/Ix; 

    output; 

    End; 

  Drop Tx i; 

Format Tx 8.0 Ex 8.2; 

Run; 

Proc Sort Data=Test4; 

   By Agegroup; 

Run; 

Creating the Report:  

To create the final life table output, we used a template from a previous SAS paper4 to create our life 
table, and then used the ODS PDF output to create the output table. We formatted the variables using 
Proc Format (not shown here).  
 
proc template; 

  define style self.border; 

 parent=styles.SansPrinter; 

 style Table / 

  rules = groups 

  frame=hsides 

       cellpadding = 3pt 

       cellspacing = 0pt 

  borderwidth = 2pt; 

      style header / 

  font_weight=bold 

  background=white 

  font_size=3; 

  end; 

run; 

 

ods listing close; 

ods pdf file="c:\temp\test.pdf" style=self.border; 

 

Proc report data=final headline headskip nowd spacing=2 split='-' center ; 

format sexRace $sexf. ; 

columns agegroup years deaths population qx Ix dx ax Lx ex; 

define agegroup /display f=$agef. "-Age-Group" width=5 center; 

define years /display "Years" width=5 center; 

define deaths /display f=comma12.0 "Number-of-Deaths" width=9 center; 



define population /display f=comma12.0 "-Estimated-Population" width=10 

center; 

define qx /display f=12.5 "(qx)" width=11 center; 

define Ix /display f=comma12.0 "(Ix)" width=11 center; 

define dx /display f=comma12.0 "(dx)" width=14 center; 

define ax /display f=8.2 "(ax)" width=6 center; 

define Lx /display f=comma12.0 "(Lx)" width=14 center; 

define ex /display f=12.2 "(ex)" width=18 center; 

 

title ="Abridged Life Tables for Texas Residents, 2014"; 

footnote =” Life expectancy at birth”; 

run; 

 

ods pdf close; 

ods listing; 

 

         Table 2: Output of the Proc Report: Abridged Life Table for Texas Residents, 2014  

 
 

 

 

 

In this way, we used SAS to produce a LE table that calculated all values as 

desired and only required minor cleanup (including adding subscripts and table 

notes) prior to publication.  

*Life expectancy at birth 
Life tables prepared using SAS 

See Technical Appendix for Life Table Construction 



Conclusion 

 
We often receive requests to “personalize” Life Expectancy table to their specific 
geographic areas, like Texas counties or by health regions. The time and resources 
involved in making these tables used to be prohibitive. The syntax created here greatly 
cuts down the time and effort to generate the life expectancy table, not only for creating 
the Texas Vital Statistics Annual Report but for all our requests. 

